

# Lecture Notes on Arrays and Ghosts

Matt Fredrikson\*

Carnegie Mellon University

Lecture 4

January 26, 2023

## 1 Introduction

At this point we have experimented with simple imperative programs over integers using loops, recursive functions, and immutable data structures. Today, we start by looking at mutable data structures, specifically arrays. How do we specify and verify simple loops that operator on arrays? The key constructs we have (loop invariants and variants) are sufficient but become more complex because they have to express not only properties of what we *change* in an array, but also about the parts that *do not change*. It is this additional requirement that makes reasoning about ephemeral (mutable) data structures in many cases more difficult than reasoning about persistent (immutable) ones.

After we understand the basics, we consider a particularly important technique to tame the basics, namely simple *models* of complex data structures. Consider, for example, a red/black tree implementation of a map. To control its complexity we have introduced the concept of a data structure invariant. In the red/black tree example, this would include the *ordering invariant* (keys in left subtrees are all smaller and keys in the right subtree are all larger than the key in a node), the *color invariant* (there are no two adjacent red nodes in the tree), and the *black height invariant* (the number of black nodes on any path from a leaf to the root is the same). These are all *internal invariants* in the sense that the implementation of the data structure must maintain them but the client should only care that red/black trees provide a correct and efficient implementation of a map from keys to values. In this case, we say the map provides a *model* of

---

\*Closely adapted from notes written by Frank Pfenning in Spring 2022

the intended behavior of the data structure. We would like the model to be *logical* and as high-level as possible to support reasoning by the client. Maps and sets are common models. In today's lecture we exemplify sets as a model of an ephemeral (mutable) implementation of bit vectors as arrays.

When implementing a data structure we have to maintain the correspondence between the low-level implementation and the high-level model. The purpose of the model is to *reason* about a data structure, but not to *compute* with it. So we would like to *erase* the code that maintains the model: actually computing it would negate all the advantages of the efficient implementation! This is the primary purpose of *ghosts*. They are pieces of code or data that exist solely for the purpose of verification and do not contribute to the outcome of the computation. This has to be checked by the verification engine. Ghost variables, or ghost fields of records, can only be used in other ghost computations. Otherwise, erasing them before the program is run would lead to incorrect code. This condition is related to the fact that *executable contracts* in C0 can not have any externally observable effects: running the program with or without executable contracts should yield the same answer (as long as all contracts are satisfied, of course).

**Learning goals.** After this lecture, you should be able to:

- Learn to formulate data structure invariants, and how work with them in Why3
- Verify simple imperative programs computing over arrays
- Model data structures using ghosts

## 2 Data Structure Invariants

As a simple example of data structure invariants we reconsider our implementation of queues. We would like to extend the interface with a function `qsize` that returns the number of elements in a queue. The obvious way to compute this would be to compute the lengths of the front and back and add them, but the complexity would be linear in the size of the queue. To answer this query in constant time we add a `size` field to the queue and maintain it as we enqueue or dequeue elements. The data structure invariant here is that the `size` field always contains the sum of the lengths of the front and back.

```
1  type queue 'a = { front : list 'a ;
2                      back : list 'a ;
3                      size : int }
4  invariant { size = length front + length back }
```

In the logic, the verifier assumes the invariant when reasoning about queues. This could create a logical inconsistency (*everything is provable!*) if the invariant is unsatisfiable. For example, the trivial invariant *false* means any program in the scope of this

declaration could now be verified. To avoid this, Why3 will try to prove that there exists an instance of the data structure for which the invariant is satisfied. For complex invariants this can be difficult, so WhyML gives you a way to specify an instance of the data structure satisfying the invariant with a 'by' clause. In this particular case it would be unnecessary, but it is good practice to always supply it.

```

1  type queue 'a = { front : list 'a ;
2                      back : list 'a ;
3                      size : int }
4  invariant { size = length front + length back }
5  by { front = Nil ; back = Nil ; size = 0 }

```

In this example it is relatively easy to update the code to maintain the size field and Why3 will prove that it is always correct. In constructing the verification condition we may think of the invariant as being *assumed* at the beginning of a function (like a precondition) and *proved* at the end of a function (like a postcondition). In between, the invariant may be violated, although this possibility does not come into play here. It is necessary because you may build or modify an element of the data structure incrementally and only at the end does the invariant hold.

The `qsize` function just returns the size field. In addition, the postcondition certifies that it is indeed the length of the queue (when viewed as a single sequence of element, which represents the client's perspective).

```

1  let qsize (q : queue 'a) : int =
2  ensures { result = length (sequence q) }
3  q.size

```

### 3 Arrays

Because of their efficiency arrays are a common data structure in imperative programs. Reasoning about arrays requires a number of techniques due to their inherent mutability and range requirements for array access.

When we modify an array as we traverse it the loop invariant generally has to be more complicated. This is because with any assignment to an array element inside a loop, we lose all information about what *any* element in the array may be. Therefore, we generally need to specify the entries of the array we change and how, and in addition that the remaining entries do not change.

As an example, we consider a function to negate every element in the set. In order to verify this, we define a predicate `negated`. As is common (and perhaps we should have done this for the `mem` predicate) test the property for a slice of the array in the interval `[lower, upper)` (inclusive the lower bound and exclusive the upper bound).

```

1  predicate negated (a : array int) (b : array int) (lower : int) (upper
   : int) =
2  a.length = b.length /\ 0 <= lower <= a.length /\ 0 <= upper <= a.
   length
3  /\ forall j:int. lower <= j < upper -> a[j] = -b[j]

```

We also ensure that the arrays  $a$  and  $b$  have the same length.

The postcondition for our negation function expresses that the state of the array upon the return is equal to the negated initial values of the array all the way up to the last element.

```

1 let negate (a: array int) : unit =
2   ensures { negated (old a) a 0 a.length}
3   let n = a.length in
4   for i = 0 to n-1 do
5     (* no variant, or invariant on i needed *)
6     invariant { ... }
7     a[i] <- -a[i]
8   done ; ()

```

This code has two additional new constructs. We use the type `unit` (whose only element is `()`) to express that the function returns no interesting value. This usually implies that it mutates some of its arguments, in this case the array  $a$ .

We also use a `for` loop, of the form `for  $i = lower$  to  $upper$  do ... done` for which we give *inclusive bounds*. It generates suitable loop invariants for the index  $lower \leq i \leq upper + 1$  and a variant of  $upper + 1 - lower$ . There is an analogous form for `for  $i = upper$  downto  $lower$  do ... done`.

What remains is to state the invariants regarding the array. Intuitively, at iteration  $i$  we have negated all the elements up to  $i$  while the elements at indices greater or equal to  $i$  have remained unchanged. To specify this we find in the useful `array.ArrayEq` module the function

```

1 array_eq_sub (a : array 'a) (b : array 'a) (lower : int) (upper : int)

```

which is true if for all  $lower \leq i < upper$  we have  $a[i] = b[i]$ . We use this where  $b$  is the original version of  $a$ .

```

1 let negate (a: array int) : unit =
2   ensures { negated (old a) a 0 a.length}
3   let n = a.length in
4   for i = 0 to n-1 do
5     (* no variant, or invariant on i needed *)
6     invariant { negated a (old a) 0 i }
7     invariant { array_eq_sub a (old a) i n }
8     a[i] <- -a[i]
9   done ; ()

```

Observe how the invariants in this form express concisely that the values in the interval  $[0, i)$  are negated, while those in the interval  $[i, n)$  are still the same as in the original array. It is generally easier to understand and express the invariant in this form than using complicated quantified formulas in-line.

It is not necessary to explicitly return the unit element (since the for-loop already returns the unit element), but we write it out for emphasis.

The code now verifies, but true to our methodology we can also try it out.

```

1 let test () =
2   let a = Array.make 4 0 in
3   ( a[0] <- 1 ; a[1] <- 27 ; a[2] <- 4 ; a[3] <- 3 ;

```

```
4 (search 1 a , search 2 a , search 27 a ) )
```

We obtain the expected answer.

```
% why3 execute intset.mlw --use="IntSet" 'test ()'
result: (int, int, int) = (0, (-1), 1)
globals:
%
```

Instead of executing it, we can also *prove* that this must be the outcome of the function.

```
1 let test () =
2 ensures { result = (0, (-1), 1) }
3 let a = Array.make 4 0 in
4 ( a[0] <- 1 ; a[1] <- 27 ; a[2] <- 4 ; a[3] <- 3 ;
5   (search 1 a , search 2 a , search 27 a ) )
```

```
% why3 prove -P alt-ergo intset.mlw
File intset-test.mlw:
Verification condition search'vc.
Prover result is: Valid (0.01s, 94 steps).
```

```
File intset-test.mlw:
Verification condition negate'vc.
Prover result is: Valid (0.03s, 383 steps).
```

```
File intset-test.mlw:
Verification condition test'vc.
Prover result is: Valid (0.03s, 160 steps).
```

## 4 Diagnosis of Failing Goals

We provide here a brief illustration how we can use the Why3 IDE to isolate failures of verification. We change the source so that the middle invariant

```
1 invariant { forall j. 0 <= j < i -> a[j] <> x }
```

has an off-by-one error

```
1 invariant { forall j. 0 <= j <= i -> a[j] <> x }
```

We start the Why3 IDE with

```
why3 ide intset.mlw
```

select the outermost goal and hit '2' for a relatively advanced strategy. It cannot prove the verification condition but splits it into several parts. We can see that many subgoals

are checked as green, but two of them still have question marks. Goals depending on them higher up in the goal/subgoal tree will then be similarly marked.

Status	Theories/Goals	Time
?	intset-bug.mlw	
?	IntSet	
?	search\vc [VC for search]	
🕒	CVC4 1.7	1.00 (steps: 270910)
🕒	Alt-Ergo 2.3.1	1.00
🕒	Z3 4.8.6	0.41 (steps: 3231242)
?	split_vc	
✅	0 [loop invariant init]	
?	1 [loop invariant init]	
✅	2 [loop invariant init]	
✅	3 [index in array bounds]	
✅	4 [loop variant decrease]	
✅	5 [loop invariant preservation]	
?	6 [loop invariant preservation]	
✅	7 [loop invariant preservation]	
✅	8 [assertion]	
✅	9 [postcondition]	
✅	10 [postcondition]	
✅	negate\vc [VC for negate]	
✅	test\vc [VC for test]	

The unproven subgoals are labeled [loop invariant init] and [loop invariant preservation] which indicates that there is a loop invariant that cannot be established initially, nor can it be shown to be preserved. To see which one we select the second one in the pane on

the left and examine the program in the pane on the right.

The screenshot shows the IDE interface for the 'intset-bug.mlw' task. On the left, the 'Theories/Goals' pane shows a tree structure under 'IntSet'. The 'split\_vc' goal is expanded, showing sub-goals 0 through 10, with goal 6 'loop invariant preservation' selected. On the right, the code editor displays the following code:

```

9 use array.ArrayEq
10
11 predicate mem (x:int) (a:array int) =
12   exists i:int. 0 <= i < a.length /\ a[i] = x
13
14 let search (x:int) (a : array int) : int =
15   ensures { (0 <= result < a.length /\ a[result] = x)
16             \/ (result = -1 /\ not (mem x a)) }
17   ensures { a = (old a) }
18   let n = a.length in
19   let ref i = 0 in
20   while i < n && a[i] <> x do
21     variant { n - i }
22     invariant { <= i <= n }
23     invariant { forall j. <= j <= i -> a[j] <> x }
24     invariant { a = (old a) }
25     i <- i + 1
26   done ;
27   assert { i = n \/ a[i] = x } ;
28   if i <> n then i else -1
29
30 predicate negated (a : array int) (b : array int) (lower : i
31   a.length = b.length /\ 0 <= lower <= a.length /\ 0 <= upper
32   /\ forall i:int. lower <= i < upper -> a[i] = -b[i]
33
34 let negate (a : array int) : unit =
35   ensures { negated (old a) a 0 a.length }
36   let n = a.length in
37   for i = 0 to n-1 do

```

The IDE highlights the assumptions `a[i] <> x` in green and the proof goal `forall j. <= j <= i -> a[j] <> x` in yellow. The message pane at the bottom shows 'The transformation made no progress'.

The IDE will highlight in green the assumptions it uses and in yellow the proof goal it is trying to prove. Here we see it is `a[j] <> x` in the middle invariant. Even though it does not occur here, the IDE will highlight formulas in your program as red if it uses its negation in the proof attempt. This occurs for the else-branch of conditionals and loop guards when reasoning about the state after the loop is exited.

Highlighting the other unprovable subgoal (labeled [loop invariant init]) leads to the same culprit. Some further forensics and thought about this will hopefully reveal the bug at this point. Fortunately, it is not in the program but in the invariant.

## 5 Loop Variants

Which postconditions arise from *loop variants*? The short description “*it has to be a non-negative quantity that strictly decreases in the loop*” is somewhat imprecise, so let’s look at the details. Why3 will check two conditions

1. If the loop guard is true, the variant  $v$  has to be nonnegative  $v \geq 0$ .
2. If we go around the loop once, the variant  $v$  strictly decreases.

The first point guarantees that if the variant is negative  $v < 0$ , the loop body will not be executed. In other words, the loop must exit if the variant becomes negative.

The first point also guarantees that the variant is nonnegative  $v \geq 0$  when entering the loop, and the second point check that it strictly decreases. Eventually, it will therefore have to become negative, at which point the loop must terminate.

As an example, consider the simple loop

```

1 let ref i = 0 in                                (* value here is irrelevant! *)
2 while (i < n) do
3   variant { n - i }
4   i <- i + 1                                    (* increment has to be positive *)
5 done

```

Here is the verification condition

```

goal f'vc :
  forall n:int.
    forall i:int.
      i < n ->
        (forall i1:int. i1 = (i + 1) -> 0 <= (n - i) /\ (n - i1) < (n - i))

```

The assumption  $i < n$  means that loop guard is true. The part  $0 \leq (n-i)$  guarantees that the variant  $n-i$  is nonnegative (which follows from  $i < n$ ). The part  $(n-i1) < (n-i)$  verifies that the variant strictly decreases, because  $i1 = i + 1$  is the value of  $i$  after one iteration of the loop. We wrote this as  $i'$  when analyzing our first mystery function.

## 6 Ghosts and Models

As an example of a model we use the standard set library to model a bit vector implementation of bounded finite sets. Here is an excerpt of the finite set module.

```

1 module Fset
2   type fset 'a
3   predicate mem (x: 'a) (s: fset 'a)
4   predicate is_empty (s: fset 'a) = forall x: 'a. not (mem x s)
5   constant empty: fset 'a
6   function add (x: 'a) (s: fset 'a) : fset 'a
7   axiom add_def: forall x: 'a, s: fset 'a, y: 'a.
8     mem y (add x s) <-> (mem y s /\ y = x)
9   function remove (x: 'a) (s: fset 'a) : fset 'a
10  axiom remove_def: forall x: 'a, s: fset 'a, y: 'a.
11    mem y (remove x s) <-> (mem y s /\ y <> x)
12  ...
13 end

```

We start by defining the Bitset module by defining a bset as a record consisting of an array `a`, a bound and a ghost field called `model` containing a finite set of integers. Because bitsets are mutable (for example, we actually *change* a bset by adding an element to it), the `model` field must also be mutable.

```

1 type bset = { a : array bool ;
2               bound : int ;
3               mutable ghost model : Fset.fset int }
4 invariant { 0 <= bound <= a.length
5             /\ forall j. 0 <= j < a.length -> a[j] <-> Fset.mem j
6             model }
6 by { a = Array.make 0 false ; bound = 0 ; model = Fset.empty }

```

The invariant states that element  $a[i]$  of the array is true if and only if the number  $i$  is in the model set. We witness the existence of such a bset with the empty array and empty model. Note that the fields `a` and `bound` are immutable, although the elements in the array are mutable.

To create an empty bset we need a bound on the elements we may add to the set, which will be the length of the array.

```
1 let empty_bset (bound : int) : bset =
2   requires { bound >= 0 }
3   ensures { Fset.is_empty result.model /\ result.bound = bound }
4   { a = Array.make bound false ; bound = bound ; model = Fset.empty }
```

The model is just the empty set. Note that the postcondition states that `empty_bset` models the empty finite set.

To add an element  $i$  to a bset we just set the corresponding array element to true (whether it was already true or not). This requires the precondition that the  $i$  is in the permissible range. Because this operation is destructive, modifying the given bset, the postcondition needs to state the model *after* the update is equal to the model *before* the update, plus the element  $i$ . For this purpose we use again the `old` keyword to refer to the state of the model at the time the function is called.

```
1 let add_bset (x : int) (s : bset) : unit =
2   requires { 0 <= x < s.bound }
3   ensures { s.model = Fset.add x (old s).model }
4   s.a[x] <- true ;
5   ghost (s.model <- Fset.add x s.model) ;
6   ()
```

The assignment to `s.model` is enclosed in `ghost (...)` to be explicit that this update of the model will not be carried out if the program is executed. Instead it is there to maintain the data structure invariant which must be restored before the function `add_bset` exits. Just before this line, by the way, the data structure invariant is false because we have updated the array but not the model. We can see the significance of checking the data structure invariance exactly at function boundaries.

Our postcondition will allow the client to reason about the effects of its add operations. Note that the pre- and post-conditions *do not reference the array*, only properties of the model and the bound. The client can reason about the behavior of these functions without knowing the representation, using only the model.

The remove operation is entirely analogous.

```
1 let remove_bset (i : int) (s : bset) : unit =
2   requires { 0 <= i < s.bound }
3   ensures { s.model = Fset.remove i (old s).model }
4   s.a[i] <- false ;
5   ghost (s.model <- Fset.remove i s.model) ;
6   ()
```

We did not implement any more complex operations such as union or intersection, even though this would certainly be possible. You can find the live-code `Bitset` module in the file [bitset.mlw](#).